Newsgroup: it.comp.lang.visual-basic
Titolo: Da un url estrarre l'indirizzo

Scusate come sapete nel mondo c'è ignoranza!
come faccio da un file .url a tirare fuori l'indirizzo(http:\\eccetera)?
ringrazio anticipatamente!

 come faccio da un file .url a tirare fuori l'indirizzo(http:\\eccetera)?
Se intendi uno shortcut URL (i file che trovi nei preferiti, ad esempio)
non sono altro che dei file di testo. La seconda riga è l'indirizzo
che cerchi ed è nel formato:
    URL=
http://eccetera/
puoi leggere il file anche con le funzioni che si usano per i file
di inizializzazione (.ini) come GetPrivateProfileString.

 come faccio da un file .url a tirare fuori l'indirizzo(http:\\eccetera)?
Non lo so ma... ti faccio un esempio di come procederei.
Ho cercato un file .URL nel disco, ho trovato, fra gli altri "pal.url". Ho
provato ad aprirlo con un editor di testo, 
p.es
. notepad, ed ecco cosa e'
venuto fuori:
[DEFAULT]
BASEURL=
http://www.comune.modena.it/cittadini/apal/html/
[DOC#6#7#8]
BASEURL=
http://www.comune.modena.it/cittadini/apal/html/mainlogo.html
ORIGURL=mainlogo.html
[DOC#6#7#9]
BASEURL=
http://www.comune.modena.it/cittadini/apal/html/index1.html
ORIGURL=index1.html
[DOC#6#10]
BASEURL=
http://www.comune.modena.it/cittadini/apal/html/main.html
ORIGURL=main.html
[InternetShortcut]
URL=
http://www.comune.modena.it/cittadini/apal/html/
Modified=C0E2453AA094BF0115
Quindi non e' altro che un file di inizializzazione (INI).
Direi quindi che puoi leggerlo con le funzioni di base per la gestione dei
file di testo (Open, Line Input, Close): ne ho postato un esempio proprio
ieri nel thread "Caricare un file di testo in una list box".
Meglio ancora, se non hai problemi, con le funzioni API per la gestione dei
file INI. Ti propongo un esempio:
Option Explicit
Private Declare Function GetPrivateProfileString _
  Lib "kernel32" _
  Alias "GetPrivateProfileStringA" ( _
    ByVal lpAppName As String _
  , ByVal lpKeyName As Any _
  , ByVal lpDefault As String _
  , ByVal lpReturnedString As String _
  , ByVal nSize As Long _
  , ByVal lpFileName As String _
  ) As Long
Public Sub Main( _
)
Const sLOC_APPNAME  As String = "InternetShortcut"
Const sLOC_FILENAME As String _
  = "C:\WINDOWS\Preferiti\VB\PAL.URL"
Const sLOC_KEYNAME  As String = "URL"
Const sLOC_DEFAULT  As String = "Assente"
Dim s_Ret As String
Dim l_Len As Long
Dim s_Msg As String
Dim i_Btn As Integer
  Let s_Ret = String$(255, vbNullChar)
  Let l_Len _
    = GetPrivateProfileString( _
          lpAppName:=sLOC_APPNAME _
        , lpKeyName:=sLOC_KEYNAME _
        , lpDefault:=sLOC_DEFAULT _
        , lpReturnedString:=s_Ret _
        , nSize:=Len(s_Ret) _
        , lpFileName:=sLOC_FILENAME)
  Let i_Btn = vbOKOnly
  If l_Len Then
    Let s_Ret = Left$(s_Ret, l_Len)
    Let s_Msg _
      = "[" & sLOC_APPNAME & "]" & vbLf _
      & sLOC_KEYNAME & "=" & vbLf
    If s_Ret = sLOC_DEFAULT Then
      Let s_Msg = s_Msg & "#" & sLOC_DEFAULT & "#"
      Let i_Btn = i_Btn + vbExclamation
    Else
      Let s_Msg = s_Msg & Chr$(34) & s_Ret & Chr$(34)
      Let i_Btn = i_Btn + vbInformation
    End If
    Call MsgBox( _
        Title:=sLOC_FILENAME _
      , Prompt:=s_Msg _
      , Buttons:=i_Btn)
  Else
    Call MsgBox( _
        Title:=App.Title _
      , Prompt:="Errore." _
      , Buttons:=i_Btn + vbCritical)
  End If
End Sub
--
Ciao.
Maurizio